Less Hot Pair Coulomb Density Matrix on a Sphere

Riccardo Fantoni*
Università di Trieste, Dipartimento di Fisica, strada Costiera 11, 34151 Grignano (Trieste), Italy
(Dated: May 31, 2026)

We tabulate the less hot diagonal pair Coulomb density matrix on a sphere and we discuss its behavior at particles contact.

Keywords: Pair Coulomb Density Matrix; Sphere; Table

CONTENTS

I. INTRODUCTION

In a quantum many body (Monte Carlo) simulation of a Coulomb liquid it is often useful to have access to the low temperature pair Coulomb density matrix ρ. In the high temperature classical limit the *primitive approximation* $\rho_{\text{primitive}}$ becomes exact. Within Path Integral Monte Carlo (PIMC) of a many body system [1] in order to access the quantum regime with less imaginary time discretization error it is important to lower the temperature in the *pair-product inter-action*. The imaginary time t is related to the temperature T of the liquid by $t=\beta=1/k_BT$ with k_B Boltzmann constant. The main ingredient to construct the pair-product inter-action is the pair density matrix. Starting from $\rho_{\text{primitive}}$ at a high temperature T_0 we can halve the temperature simply by squaring it [2, 3] (see also section IV.G of Ref. [1]). This lower temperature pair density matrix already possesses different properties than the primitive approximation. In position representation, on the diagonal

$$\left|\frac{\phi(\boldsymbol{r}_1,\boldsymbol{r}_2)}{k_BT_0}\right|=\left|\ln\left[\frac{\rho_{\text{primitive}}(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;1/k_BT_0)}{\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;1/k_BT_0)}\right]\right|\xrightarrow{\boldsymbol{r}_1\to\boldsymbol{r}_2}\infty, \tag{1.1}$$

where $\phi(\boldsymbol{r}_1,\boldsymbol{r}_2)$ is the pair Coulomb potential between particle 1 at $\boldsymbol{r}_1$ and particle 2 at $\boldsymbol{r}_2$. On the other hand we can reach the temperature $T_0=T_n/n$ through the *continuum limit* sending $n\to\infty$ and $T_n\to\infty$. In this case we find

$$|P(\boldsymbol{r}_1,\boldsymbol{r}_2;1/k_BT_0)|=\lim_{n\to\infty}\left|\ln\left[\frac{[\rho_{\text{primitive}}]^n(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;n/k_BT_n)}{[\rho_0]^n(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;n/k_BT_n)}\right]\right|\xrightarrow{\boldsymbol{r}_1\to\boldsymbol{r}_2}\text{finite}, \tag{1.2}$$

where we denote with $[\rho]^n$ the nth power of the pair density matrix. Moreover a *cusp condition* must hold for P as the two particles approach each other [1, 4, 5].

* riccardo.fantoni@scuola.istruzione.it

The same contact value smearing property holds at any temperature T_0. This feature makes it of fundamental importance not to use the primitive approximation for example in an electron proton binary mixture where the divergence of the attractive Coulomb potential between unlike species at contact would inevitably let the fast electrons stick onto the slow protons. In other words quantum statistical physics does not need hard core regularization which is essential in classical statistical physics instead.

In this work we will show how this can be accomplished on a Sphere. We will not give the exact solution for the pair Coulomb density matrix as was done in flat space in Refs. [5–7] but we will give an approximate numerical solution instead. If we were able to find an exact solution of the pair Coulomb density matrix on a sphere we would not need a method to lower the temperature from the classical high temperature limit primitive approximation. Moreover the squaring method can be used only if either an exact solution is known, so that successive squaring allow to reach $T_0/2^n$ temperatures, or we start from very high temperature T_0 and we just want to halve the temperature once. Here we propose to use the path integral discretized in imaginary time with timestep $\tau = 1/k_BT_n$ to reach the temperatures $T_0 = 1/nk_B\tau$ approaching the continuum limit $\tau \to 0$.

We will only worry about the *end-point approximation* (see section IV.F of Ref. [1]) to the pair Coulomb density matrix. Even if this is exact on a flat space due to the peculiar properties of the Coulomb pair potential [5] it should be regarded just as an approximation on the Sphere.

II. LOWER TEMPERATURE PAIR COULOMB DENSITY MATRIX

We show a method to lower the temperature of the hot Pair Coulomb Density Matrix (PCDM) on a Sphere so to extend the work in flat space of Refs. [5–7] to a curved surface.

We will denote with a Greek index a coordinate component and with a Roman index a particle label. We will adopt Einstein summation convention to tacitly assume a sum over repeated Greek indexes. We distinguish between upper contravariant Greek indexes and lower covariant Greek indexes. Passing between the two is accomplished with a contraction with the metric tensor, as usual.

The metric tensor of the Sphere of radius a is

$$||g_{\alpha\beta}|| = a^2 \begin{pmatrix} 1 & 0 \\ 0 & \cos^2\theta \end{pmatrix}. \tag{2.1}$$

with $g(\boldsymbol{r}) = \det||g_{\alpha\beta}|| = a^4\cos^2\theta$ and $\theta \in [-\pi/2, \pi/2]$ the polar angle measured from the equatorial plane, so that $\theta = \pi/2$ is the north pole and $\theta = -\pi/2$ is the south pole.

It is important to realize that on a curved surface the two body problem is generally not separable into a center of mass and a relative one body problem. Therefore, in full generality, we will study the position representation $\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t)$ of the density matrix $\rho = \exp(-Ht)$, where H is the Hamiltonian operator for the two bodies an t is the imaginary time that in statistical physics is identified with the inverse temperature $\beta = 1/k_BT$ with k_B Boltzmann constant. These are initially at $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$ and at an imaginary time t later they are at $\boldsymbol{r}'_1$ and $\boldsymbol{r}'_2$.

Therefore, the PCDM on the Sphere will satisfy the Bloch equation

$$\frac{\partial\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t)}{\partial t} = \left(\frac{\lambda_1}{\sqrt{g(\boldsymbol{r}_1)}}\nabla_{1\alpha}\sqrt{g(\boldsymbol{r}_1)}\nabla_1^\alpha + \frac{\lambda_2}{\sqrt{g(\boldsymbol{r}_2)}}\nabla_{2\alpha}\sqrt{g(\boldsymbol{r}_2)}\nabla_2^\alpha - \phi(\boldsymbol{r}_1, \boldsymbol{r}_2)\right)\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t), \tag{2.2}$$

$$\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; 0) = \delta(\boldsymbol{r}_1 - \boldsymbol{r}'_1)\delta(\boldsymbol{r}_2 - \boldsymbol{r}'_2)/\sqrt{g(\boldsymbol{r}_1)g(\boldsymbol{r}_2)}, \tag{2.3}$$

where, for particle $i = 1, 2$, $\lambda_i = \hbar^2/2m_i$ with m_i the mass of particle i, $g(\boldsymbol{r}_i) = a^4\sin^2\theta_i$, $\boldsymbol{\nabla}_i = (\nabla_{i1}, \nabla_{i2}) = (\partial/\partial\theta_i, \partial/\partial\varphi_i)$ is the gradient respect to the coordinates $\boldsymbol{r}_i = (r_i^1, r_i^2) = (\theta_i, \varphi_i)$, δ is a two dimensional Dirac delta function, ρ is the pair density matrix, and ϕ is the Coulomb pair potential.

The Coulomb pair potential for particles moving **on** the surface can be chosen as

$$\phi_{\rm on}(\boldsymbol{r}_1, \boldsymbol{r}_2) = Z_1Z_2e^2/d(\boldsymbol{r}_1, \boldsymbol{r}_2), \tag{2.4}$$

where Z_ie is the charge of particle i and for the distance between the two points $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$, $d(\boldsymbol{r}_1, \boldsymbol{r}_2)$, on the Sphere we can either choose the geodesic distance

$$d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) = a\arccos[\sin\theta_1\sin\theta_2 + \cos\theta_1\cos\theta_2\cos(\varphi_1 - \varphi_2)], \tag{2.5}$$

or, viewing the Sphere as embedded in the three dimensional space, the Euclidean distance

$$d_e(\boldsymbol{r}_1, \boldsymbol{r}_2) = 2a\sin(d_g/2a). \tag{2.6}$$

For particles living **in** the surface [8] we may choose

$$\phi_{\rm in}(\boldsymbol{r}_1,\boldsymbol{r}_2)=-Z_1Z_2e^2\ln[d_e(\boldsymbol{r}_1,\boldsymbol{r}_2)/\ell], \tag{2.7}$$

with ℓ a constant with the dimensions of a distance. We will call $\phi_{\rm on}^g$, $\phi_{\rm on}^e$, $\phi_{\rm in}$ the three cases just illustrated.

We will measure energy in units of the Hartree e^2/a_B, with e the charge of the electron and $a_B=\hbar^2/m_ee^2$ Bohr radius, with m_e the mass of the electron. And length in units of the Bohr radius a_B. We will call $\mu_i=m_i/m_e$ the reduced mass of particle i. Temperatures are in units of $\hbar^2/m_ea_B^2k_B$.

The free particles, $\phi\to 0$, pair density matrix can be determined from the work of Bastianelli and Corradini [9]. We will call their solution for the single free particle i, $\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i';t)$. Therefore the pair free density matrix will be

$$\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)=\rho_{10}(\boldsymbol{r}_1|\boldsymbol{r}_1';t)\rho_{20}(\boldsymbol{r}_2|\boldsymbol{r}_2';t). \tag{2.8}$$

Then our PCDM can be written as

$$\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)=\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)e^{-P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)}, \tag{2.9}$$

where P is the *inter-action* [1].

Aim of this work is to tabulate $P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)$ for the three Coulomb cases illustrated above, $P_{\rm on}^g$, $P_{\rm on}^e$, $P_{\rm in}$. In the hot case, for small $t\ll 1$, for each of these cases, the *primitive approximation* [1]

$$P_{\rm primitive}\approx t\phi, \tag{2.10}$$

holds.

Therefore, introducing a timestep τ, we can find the PDCM at all subsequent timesteps through the following convolution path integral

$$\begin{aligned}\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';n\tau)=\int&\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_{1,1},\boldsymbol{r}_{2,1};\tau)\rho(\boldsymbol{r}_{1,1},\boldsymbol{r}_{2,1}|\boldsymbol{r}_{1,2},\boldsymbol{r}_{2,2};\tau)\ldots\rho(\boldsymbol{r}_{1,n-1},\boldsymbol{r}_{2,n-1}|\boldsymbol{r}_1',\boldsymbol{r}_2';\tau)\times\\&\prod_{k=1}^{n-1}\sqrt{g(\boldsymbol{r}_{1,k})g(\boldsymbol{r}_{2,k})}d\boldsymbol{r}_{1,k}\boldsymbol{r}_{2,k},\end{aligned} \tag{2.11}$$

where we denote with $\boldsymbol{r}_{i,k}$ the position of particle i at timeslice k, $d\boldsymbol{r}=d\theta d\varphi$, and we choose τ small enough so that

$$\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';\tau)\approx\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';\tau)e^{-\tau[\phi(\boldsymbol{r}_1,\boldsymbol{r}_2)+\phi(\boldsymbol{r}_1',\boldsymbol{r}_2')]/2} \tag{2.12}$$

$$=\rho_{10}(\boldsymbol{r}_1|\boldsymbol{r}_1';\tau)\rho_{20}(\boldsymbol{r}_2|\boldsymbol{r}_2';\tau)e^{-\tau[\phi(\boldsymbol{r}_1,\boldsymbol{r}_2)+\phi(\boldsymbol{r}_1',\boldsymbol{r}_2')]/2}, \tag{2.13}$$

is a good approximation. We will perform the $4(n-1)$ integrations in Eq. (2.11) through a simple quadrature method.

According to the analysis of Ref. [9] we can further approximate

$$g^{1/4}(\boldsymbol{r}_i)\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i';\tau)g^{1/4}(\boldsymbol{r}_i')\approx\frac{\mu_i}{2\pi\tau}\exp\left\{-\frac{\mu_i}{2\tau}\delta_{\alpha\beta}(\boldsymbol{r}_i-\boldsymbol{r}_i')^\alpha(\boldsymbol{r}_i-\boldsymbol{r}_i')^\beta-\tau\frac{V_R(x_i)+V_R(x_i')}{2}\right\}, \tag{2.14}$$

where $\delta_{\alpha\beta}$ is the Kronecker delta, the normalization $2\pi\tau/\mu_i$ is just the one of a free particle of mass μ_i on a plane and corresponds to the exact path integral performed with an action $\int_0^\tau dt'\,\mu_i\delta_{\alpha\beta}\dot{\boldsymbol{r}}_i^\alpha\dot{\boldsymbol{r}}_i^\beta/2$, $x_i^2=\delta_{\alpha\beta}{\boldsymbol{r}_i}^\alpha{\boldsymbol{r}_i}^\beta$, and V_R is the effective potential due to the curvature of the Sphere, $R=2M^2$ with $M=1/a$, namely

$$V_R(x)=-\frac{M^2}{6}-\frac{5(Mx)^2-3+[(Mx)^2+3]\cos(2Mx)}{48x^2\sin^2(Mx)}, \tag{2.15}$$

which is equation (2.32) of Ref. [9].

We may additionally introduce the following exact Feynman-Kac expression for the one free body density matrix at time t

$$g^{1/4}(\boldsymbol{r}_i)\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i';t)g^{1/4}(\boldsymbol{r}_i')=\left\langle e^{-S_R}\right\rangle,\quad i=1,2 \tag{2.16}$$

$$S_R=\int_0^t dt'\,V_R(x_i) \tag{2.17}$$

where $\langle\ldots\rangle$ denotes an average over all single particle Gaussian random walks from $\boldsymbol{r}_i$ to $\boldsymbol{r}_i'$ in a time t. In Appendix A we give the diagonal part $\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i;t)=\rho_{i0}(\mathbf{0}|\mathbf{0};t)$ up to orders t^8. Our Eq. (A1) coincides with the perturbative expansion (3.23) of Ref. [9] for $d=2$ spatial dimensions.

Even if in general we will be interested in the full $P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)$ in practice in a many body simulation we may use the *end point approximation* for the inter-action as shown in section IV.F of Ref. [1] where one only needs the diagonal components $P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t)=P(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)$. Where on a sphere the geodesic distance $d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)$ between any two points $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$ is bounded from above by πa. Due to the Sphere symmetry, without loss of generality, we may choose $\boldsymbol{r}_1=(\pi/2,0)$ and $\boldsymbol{r}_2=(\theta,0)$, so that $d_g=a(\pi/2-\theta)$.

On the diagonal we will have

$$\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t)=\rho(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)=\rho_0(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)e^{-P(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)}, \tag{2.18}$$

we will then tabulate

$$P(m\delta;n\tau)=-\ln\left[\frac{\rho(m\delta;n\tau)}{\rho_0(m\delta;n\tau)}\right],\quad n=1,2,3,\ldots,\quad m=1,2,3\ldots,m_{\max} \tag{2.19}$$

for given small τ and δ so that $m_{\max}=\pi/\delta$. For an imaginary time $t_n=n\tau$ this will require a number of $4(n-1)$ quadratures for each m to calculate $\rho(m\delta;n\tau)=\rho(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t_n)=\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t_n)$ and $\rho_0(m\delta;n\tau)=\rho_0(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t_n)=\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t_n)$ for $m=1,2,3,\ldots,m_{\max}$, through the corresponding path integral convolution (2.11). Each convolution requires 4 quadratures over the azimuthal angle $\varphi\in[-\pi,\pi[$ and the polar angle $\theta\in[-\pi/2,\pi/2]$ of each particle. Choosing an angular discretization $\delta_\varphi=2\delta_\theta$ we will have that each angular integral reduces to a sum over $l=\pi/\delta_\theta$ terms. Then in order to tabulate $P(d_g/a;t_n)$ we need $l^{4(n-1)}$ functions evaluations for each m. This is exponential in n so is only useful for small n. The squaring of the primitive approximation to the PCDM is obtained for $n=2$ so that $t=2\tau$. Already for $n=3$ the calculation becomes extremely computationally demanding at large l. In any case, in a many body simulation the pair-product inter-action is needed at a fixed small t, which determines the discretization error [1]. So one needs to tabulate the PDCM just once, at the beginning of the simulation.

In Table I we show $P(m\delta;n\tau)$ for $a=1,\mu_1=\mu_2=1,Z_1=-Z_2=-1$, $\tau=0.1,\delta=0.3$ and $n=3$ and compare it to the primitive approximation $n\tau\phi(m\delta)$.

In Fig. 1 we plot the ratio $R(m\delta)=P(m\delta;n\tau)/n\tau\phi(m\delta)$ as a function of m for $a=1,\mu_1=\mu_2=1,Z_1=-Z_2=1$, $\tau=0.01,\delta=0.03$ and $n=2,l=30$ and $n=3,l=10$. We see how at large particles distance the ratio approaches 1 but at contact it tends to $\approx 1/n$. Therefore we expect that at constant $t=n\tau$ as $n\to\infty$, in the $\tau\to 0$ continuum limit, the ratio $R(0)\to 0$. Moreover the PCDM at particles contact must obey to a cusp condition [5] as discussed in the next section.

TABLE I. Table for $P(m\delta;n\tau)$ for $a=1,\mu_1=\mu_2=1,Z_1=-Z_2=1$, $\tau=0.1,\delta=0.3$ and $n=3$. For the angular convolutions in Eq. (2.11) we used a simple quadrature method with $l=5$. For comparison we also report the values of the Coulomb potential $n\tau\phi(m\delta)$.

m	$P_{\rm in}$	$t\phi_{\rm in}$	$P^g_{\rm on}$	$t\phi^g_{\rm on}$	$P^e_{\rm on}$	$t\phi^e_{\rm on}$
1	$-0.217D+00$	$-0.362D+00$	$-0.652D+00$	$-0.100D+01$	$-0.658D+00$	$-0.100D+01$
2	$-0.149D+00$	$-0.158D+00$	$-0.485D+00$	$-0.500D+00$	$-0.493D+00$	$-0.508D+00$
3	$-0.110D+00$	$-0.418D-01$	$-0.429D+00$	$-0.333D+00$	$-0.438D+00$	$-0.345D+00$
4	$-0.814D-01$	$0.365D-01$	$-0.398D+00$	$-0.250D+00$	$-0.409D+00$	$-0.266D+00$
5	$-0.262D-01$	$0.930D-01$	$-0.337D+00$	$-0.200D+00$	$-0.351D+00$	$-0.220D+00$
6	$0.677D-01$	$0.135D+00$	$-0.227D+00$	$-0.167D+00$	$-0.246D+00$	$-0.191D+00$
7	$0.103D+00$	$0.165D+00$	$-0.194D+00$	$-0.143D+00$	$-0.217D+00$	$-0.173D+00$
8	$0.152D+00$	$0.187D+00$	$-0.153D+00$	$-0.125D+00$	$-0.182D+00$	$-0.161D+00$
9	$0.169D+00$	$0.201D+00$	$-0.138D+00$	$-0.111D+00$	$-0.172D+00$	$-0.154D+00$
10	$0.193D+00$	$0.207D+00$	$-0.116D+00$	$-0.100D+00$	$-0.158D+00$	$-0.150D+00$

In Appendix B we present the FORTRAN code used to generate the Table I and Fig. 1.

III. CUSP CONDITION

On the sphere the cusp condition [1, 4, 5]. can be determined studying the asymptotic behavior as $\boldsymbol{r}_1\to\boldsymbol{r}_2$ of a vanishing residual energy for the pair-product approximation to the inter-action [1]. On a Sphere as $\boldsymbol{r}_1\to\boldsymbol{r}_2$ we should compare

$$\left(\frac{\lambda_1}{\sqrt{g(\boldsymbol{r}_1)}}\nabla_{1\alpha}\sqrt{g(\boldsymbol{r}_1)}\nabla_1^\alpha+\frac{\lambda_2}{\sqrt{g(\boldsymbol{r}_2)}}\nabla_{2\alpha}\sqrt{g(\boldsymbol{r}_2)}\nabla_2^\alpha\right)P(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2);t)=-\phi(\boldsymbol{r}_1,\boldsymbol{r}_2). \tag{3.1}$$



FIG. 1. We show the plot of the ratio $R(m\delta) = P^g_{\rm on}(m\delta; n\tau)/n\tau\phi^g_{\rm on}(m\delta)$ as a function of $m\delta$ for $a = 1, \mu_1 = \mu_2 = 1, Z_1 = -Z_2 = 1$, $\tau = 0.01, \delta = 0.03$ and $n = 2, l = 30$ and $n = 3, l = 10$. The oscillations observed for $l = 10$ are a numerical artifact of the error in the calculation of the angular convolutions. From the case $n = 2$ we see how already for $l = 30$ the squaring is close to convergence. But for $n = 3$ the case $l = 30$ is prohibitively computationally demanding.

As $\boldsymbol{r}_1 \to \boldsymbol{r}_2$, $d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) \to 0$ and we are looking at a local property on the sphere so that we can use the result on the tangent plane at the point $\boldsymbol{r}_1 = \boldsymbol{r}_2$ namely

$$\lim_{d_g\to 0} \frac{\partial P_{\rm on}(d_g;t)}{\partial d_g} = \frac{Z_1 Z_2}{\lambda_1 + \lambda_2}, \tag{3.2}$$

$$\lim_{d_g\to 0} \frac{\partial P_{\rm in}(d_g;t)}{\partial d_g} = 0, \tag{3.3}$$

where we used the following expression for the Laplacian in an Euclidean space of dimension d in an isotropic state

$$\boldsymbol{\nabla}^2_r = \frac{\partial^2}{\partial r^2} + \frac{d-1}{r}\frac{\partial}{\partial r}. \tag{3.4}$$

IV. CONCLUSIONS

We tabulated the pair Coulomb density matrix on the Sphere obtained by lowering the temperature of the classical high temperature primitive approximation through a discretized path integral calculation. As expected we found that the inter-action does not diverge at particles contact as occurs in the primitive approximation. Since we decided to calculate the discretized path integral convolutions with simple quadratures we are constrained to keep far from the continuum limit $\tau \to 0$. This can be approached efficiently with Monte Carlo integrations [10] instead.

We also derive the cusp condition for the pair Coulomb density matrix on the sphere.

Appendix A: $\rho_{i0}(\mathbf{0},\mathbf{0};t)$

Calling $\mu_i = m_i/m_e$ and $R = 2/a^2$ the Sphere scalar curvature, we find from equation (3.23) of Ref. [9]

$$\ln\left[2\pi t\sqrt{g(\mathbf{0})}\rho_{i0}(\mathbf{0}|\mathbf{0};t)/\mu_i\right] = \frac{tR}{12} + \frac{(tR)^2}{6!}\frac{1}{2} + \frac{(tR)^3}{9!}16 + \frac{(tR)^4}{10!}\frac{59}{4} + \frac{(tR)^5}{11!}\frac{58}{3} + \frac{(tR)^6}{13!}\frac{550789}{1260} + \frac{(tR)^7}{14!}\frac{46838}{45} + \frac{(tR)^8}{17!}\frac{596004707}{720} + \mathscr{O}(t^9). \tag{A1}$$

Appendix B: The code

We give here the FORTRAN code listing to generate the table of Eq. (2.19):

```
      program PCDMS
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
c
c     Hot Pair Coulomb Density Matrix for End-Point Approximation
c     on a Sphere
c
c     0<ph<2*pi   -pi/2<th<pi/2
c     (ds/a)**2 = dph**2 + [cos(th)*dth]**2
c
c     a     = Sphere radius
c     tau   = time step
c     delta = geodesic distance step
c     da    = angle discretization
c
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
      implicit none
      integer*8 mni
      real*8    a,tau,delta,eps,pi
      parameter(mni=1000)
      parameter(a=1.,tau=0.01,delta=0.03,eps=1.d-6)
      real*8    mu1,mu2,z1,z2,ti,da,dth,dph
      real*8    th10,ph10,th20,ph20,dg0,de0
      real*8    th1,ph1,th2,ph2,dg,de
      real*8    th1p,ph1p,th2p,ph2p,dgp,dep
      real*8    rhoid,rho,pot,pcdm,fa,pidm,faid
      integer*8 i,j,k,ni,na,id,nd
      integer*8 ith1(mni),iph1(mni),ith2(mni),iph2(mni)
      character potk*3,tabf*11

      pi=acos(-1.d0)

      write(*,*) '..................................................'
      write(*,*) 'Sphere radius       a =',a
      write(*,*) 'time step         tau =',tau
      write(*,*) 'g distance step delta =',delta
      write(*,*) '..................................................'
      write(*,*) 'mu1 and mu2 in electron mass units'
      read (*,*) mu1,mu2
      write(*,*) 'Z1 and Z2'
      read (*,*) z1,z2
      write(*,*) '# timestep iterations, n >= 1, n <=',mni
      read (*,*) ni
      ti =(ni+1)*tau
      write(*,*) 'imaginary time t = (n+1)*tau =',ti
      write(*,*) '# angular distance discretization points, na'
      read (*,*) na
      da =pi/na
      write(*,*) 'angle discretization step da = pi/na =',da
      write(*,*) 'give the kind of Coulomb potential (in,one,ong)'
      read (*,*) potk
      tabf = 'tab-'//trim(potk)//'.dat'
      open(unit=8,file=tabf,status='unknown')

      dth = da
      dph = 2*da
      nd  = int(pi/delta)

      ith1=1
      ith2=1
      iph1=1
```

```
      iph2=1
c     loop over geodesic distance
      do id=1,nd
         th10=id*delta-pi/2
         ph10=eps
         th20=-pi/2
         ph20=eps
         dg0=a*acos(sin(th10)*sin(th20)+cos(th10)*cos(th20)*
     &        cos(ph10-ph20))
         de0=2*a*sin(dg0/2/a)
         pcdm = 0.
         pidm = 0.
c     loop over timesteps
 10      th1=ith1(1)*dth-pi/2
         ph1=iph1(1)*dph
         th2=ith2(1)*dth-pi/2
         ph2=iph2(1)*dph
         dg=a*acos(sin(th1)*sin(th2)+cos(th1)*cos(th2)*cos(ph1-ph2))
         de=2*a*sin(dg/2/a)
         if(dg.lt.eps)goto 20
         fa=rho(potk,tau,a,mu1,mu2,z1,z2,
     &        th10,ph10,th20,ph20,th1,ph1,th2,ph2,dg,de,dg0,de0)
         faid=rhoid(tau,a,mu1,mu2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20)
         do j=2,ni
            th1=ith1(j-1)*dth-pi/2
            ph1=iph1(j-1)*dph
            th2=ith2(j-1)*dth-pi/2
            ph2=iph2(j-1)*dph
            th1p=ith1(j)*dth-pi/2
            ph1p=iph1(j)*dph
            th2p=ith2(j)*dth-pi/2
            ph2p=iph2(j)*dph
            dg=a*acos(sin(th1)*sin(th2)+cos(th1)*cos(th2)*cos(ph1-ph2))
            de=2*a*sin(dg/2/a)
            dgp=a*acos(sin(th1p)*sin(th2p)+cos(th1p)*cos(th2p)*
     &           cos(ph1p-ph2p))
            dep=2*a*sin(dgp/2/a)
            if(dg .lt.eps)goto 20
            if(dgp.lt.eps)goto 20
            fa=fa*dth**2.*dph**2.*rho(potk,tau,a,mu1,mu2,z1,z2,
     &           th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p,dg,de,dgp,dep)
            faid=faid*dth**2.*dph**2.*rhoid(tau,a,mu1,mu2,
     &           th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p)
         enddo
         th1=ith1(ni)*dth-pi/2
         ph1=iph1(ni)*dph
         th2=ith2(ni)*dth-pi/2
         ph2=iph2(ni)*dph
         dg=a*acos(sin(th1)*sin(th2)+cos(th1)*cos(th2)*
     &        cos(ph1-ph2))
         de=2*a*sin(dg/2/a)
         if(dg.lt.eps)goto 20
         fa=fa*dth**2.*dph**2.*rho(potk,tau,a,mu1,mu2,z1,z2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20,dg,de,dg0,de0)
         faid=faid*dth**2.*dph**2.*rhoid(tau,a,mu1,mu2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20)
         pcdm = pcdm+fa
         pidm = pidm+faid
 20      do k=1,ni
            ith1(k)=ith1(k)+1
            if(ith1(k).le.na) goto 10
            ith2(k)=ith2(k)+1
            ith1(k)=1
            if(ith2(k).le.na) goto 10
            iph1(k)=iph1(k)+1
            ith1(k)=1
            ith2(k)=1
            if(iph1(k).le.na) goto 10
            iph2(k)=iph2(k)+1
            ith1(k)=1
            ith2(k)=1
            iph1(k)=1
            if(iph2(k).le.na) goto 10
            do j=1,k
               ith1(j)=1
               ith2(j)=1
               iph1(j)=1
               iph2(j)=1
            enddo
         enddo
c     tabulate P=-log(rho/rho0)
         pcdm = -log(pcdm/pidm)
         write(8,100) a*dg0,pcdm,ti*pot(potk,a,z1,z2,dg0,de0)
         write(*,*)   a*dg0,pcdm,ti*pot(potk,a,z1,z2,dg0,de0),
     &        pcdm/(ti*pot(potk,a,z1,z2,dg0,de0))
      enddo

 100  format(3(d10.3,3x))
      close(unit=8)

      stop
      end


      function fact(n)
      implicit none
c     factorial function
      real*8    fact,p
      integer   i,n

      p=1.
      do i=1,n
         p=p*dble(i)
      enddo
      fact=p

      return
      end
```

```
      function vr(a,mu,th,ph)
      implicit none
c     Eq. (2.32) Bastianelli and Corradini EPJC 77, 731 (2017)
      real*8    vr,a,mu,th,ph
      real*8    pi,m,x,mx,eps
      eps=1.d-1
      pi=acos(-1.d0)

      m =1/a
      x =sqrt(th**2.+ph**2.)
      mx=m*x
      vr=-m**2./6-(5*mx**2.-3+(mx**2.+3)*cos(2*mx))/48/x**2./sin(mx)**2.
      if(abs(vr).gt.1/eps) vr=0.

      return
      end


      function pot(potk,a,z1,z2,dg,de)
      implicit none
c     the Coulomb potential
      real*8    pot,a,z1,z2
      real*8    pi,de,dg
      character potk*3
      pi=acos(-1.d0)

      if(potk.eq.'in') then
         pot=-z1*z2*log(de)
      elseif(potk.eq.'one') then
         pot=z1*z2/de
      elseif(potk.eq.'ong') then
         pot=z1*z2/dg
      else
         write(*,*) 'no pair-potential available !'
         stop
      endif

      return
      end


      function rho(potk,tau,a,mu1,mu2,z1,z2,
     &     th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p,dg,de,dgp,dep)
      implicit none
c     primitive approximation for rho
      real*8    rho,tau,a,mu1,mu2,z1,z2,dg,de,dgp,dep
      real*8    th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p
      real*8    pot,vr,ddx1,ddx2
      character potk*3

      ddx1=(th1-th1p)**2.+(ph1-ph1p)**2.
      ddx2=(th2-th2p)**2.+(ph2-ph2p)**2.

      rho =    (mu1/2/tau)*exp(-ddx1/2/mu1/tau-tau*
     &     (vr(a,mu1,th1,ph1)+vr(a,mu1,th1p,ph1p))/2)
      rho =rho*(mu2/2/tau)*exp(-ddx2/2/mu2/tau-tau*
     &     (vr(a,mu2,th2,ph2)+vr(a,mu2,th2p,ph2p))/2)
      rho =rho*exp(-tau*(pot(potk,a,z1,z2,dg,de)+
     &     pot(potk,a,z1,z2,dgp,dep))/2)

      return
      end


      function rhoid(tau,a,mu1,mu2,
     &     th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p)
      implicit none
c     primitive approximation for rho
      real*8    rhoid,tau,a,mu1,mu2
      real*8    th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p
      real*8    vr,ddx1,ddx2

      ddx1=(th1-th1p)**2.+(ph1-ph1p)**2.
      ddx2=(th2-th2p)**2.+(ph2-ph2p)**2.

      rhoid =      (mu1/2/tau)*exp(-ddx1/2/mu1/tau-tau*
     &     (vr(a,mu1,th1,ph1)+vr(a,mu1,th1p,ph1p))/2)
      rhoid =rhoid*(mu2/2/tau)*exp(-ddx2/2/mu2/tau-tau*
     &     (vr(a,mu2,th2,ph2)+vr(a,mu2,th2p,ph2p))/2)

      return
      end
```

AUTHOR DECLARATIONS

Conflicts of interest

None declared.

Data availability

The data that support the findings of this study are available from the corresponding author upon reasonable request.

Funding

None declared.

[1] D. M. Ceperley, Path integrals in the theory of condensed Helium, Rev. Mod. Phys. **67**, 279 (1995).
[2] R. G. Storer, Path-Integral Calculation of the Quantum-Statistical Density Matrix for Attractive Coulomb Forces, J. Math. Phys. **9**, 964 (1968).
[3] A. D. Klemm and R. G. Storer, The structure of quantum fluids: helium and neon, Aust. J. Phys. **26**, 43 (1973).
[4] J. C. Kimball, Short-Range Correlations and Electron-Gas Response Functions, Phys. Rev. A **7**, 1648 (1973).
[5] E. L. Pollock, Properties and computation of the Coulomb pair density matrix, Comput. Phys. Commun. **52**, 49 (1988).
[6] P. Vieillefosse, Coulomb Pair Density Matrix I, J. Stat. Phys. **74**, 1195 (1994).
[7] P. Vieillefosse, Coulomb Pair Density Matrix II, J. Stat. Phys. **80**, 461 (1995).
[8] R. Fantoni, Plasma living in a curved surface at some special temperature, Physica A **177**, 524 (2019).
[9] F. Bastianelli and O. Corradini, On the simplified path integral on spheres, Eur. Phys. J. C **77**, 731 (2017).
[10] E. L. Pollock and D. M. Ceperley, Simulation of quantum many-body systems by path-integral methods, Phys. Rev. B **30**, 2555 (1984).